Key Homework 18

Math 277, Fall 2005

Ordinary Differential Equations

Dr. Goutziers

Initializations

```
> restart;
```

3: Page 186

Observe that

$$\frac{d}{dt}3^t = 3^t \ln 3, \frac{d^2}{dt^2}3^t = 3^t \ln^2 3, \cdots$$

Hence, the differential family of 3^t equals $\{3^t\}$. Since it is finite, the method of undetermined coefficients can be applied to the differential equation

$$x'' + 5x' - 3x = 3^t$$

4: Page 186

Observe that

$$\frac{d}{dx}\left[e^{-4x}\sin x\right] = e^{-4x}\left(\cos x - 4\sin x\right), \frac{d^2}{dx^2}\left[e^{-4x}\sin x\right] = e^{-4x}\left(15\sin x - 8\cos x\right), \cdots$$

Hence, the differential family of $\mathbf{e}^{(-4x)}\sin(x)$ equals $\{\mathbf{e}^{(-4x)}\sin(x), \mathbf{e}^{(-4x)}\cos(x)\}$. Since it is finite, the method of undetermined coefficients can be applied to the differential equation

$$2y''(x) - 6y'(x) + y(x) = \mathbf{e}^{(-4x)}\sin(x)$$

5: Page 186

Observe that

$$\frac{d}{d\theta}\sec\theta = \sec\theta\tan\theta, \frac{d^2}{d\theta^2}\sec\theta = \sec\theta\tan^2\theta - \sec^3\theta$$

So the differential family of $\sec(\theta)$ appears not to be finite and the method of undetermined coefficients cannot be applied to the differential equation

$$y''(\theta) + 3y'(\theta) - y(\theta) = \sec(\theta)$$

10: Page 186

Code the differential equation.

```
> deq:=diff(y(t), t$2)+2*diff(y(t), t)-y(t)=10;
```

$$deq := \left(\frac{d^2}{dt^2}y(t)\right) + 2\left(\frac{d}{dt}y(t)\right) - y(t) = 10$$

The differential family of 10 is $\{1\}$. Moreover, it is easy to see that $y(t) = 1$ it not a solution to the homogeneous equation, so we use a trial solution of the form

$$y_{try}(t) = c_1$$

```
> ytry:=c[1];
```

$$ytry := c_1$$

Substitute this trial solution into the differential equation and solve for c_1.

```
> eq_p:=eval(subs(y(t)=ytry, deq));
```

```
val_p:=isolate(eq_p, c[1]);
```

$$eq\_p := -c_1 = 10$$

$$val\_p := c_1 = -10$$

A particular solution to this differential equation is given by

```
> yp:=subs(val_p, ytry);
```

$$yp := -10$$

```
>
```

13: Page 186

Code the differential equation.

```
> deq:=diff(y(t), t$2)-diff(y(t), t)+9*y(t)=3*sin(3*t);
```

$$deq := \left(\frac{d^2}{dt^2}\mathrm{y}(t)\right) - \left(\frac{d}{dt}\mathrm{y}(t)\right) + 9\,\mathrm{y}(t) = 3\sin(3\,t)$$

The differential family of $3\sin(3\,t)$ is $\{\sin(3\,t), \cos(3\,t)\}$. Moreover, because the coefficient of y' is not zero, $\sin(3\,t)$ is not a solution to the homogeneous equation, so we use a trial solution of the form

$$y_{try}(t) = c_1\sin(3\,t) + c_2\cos(3\,t)$$

```
> ytry:=c[1]*sin(3*t)+c[2]*cos(3*t);
```

$$ytry := c_1\sin(3\,t) + c_2\cos(3\,t)$$

Substitute this trial solution into the differential equation and solve for c_1 and c_2.

```
> eq_p:=eval(subs(y(t)=ytry, deq));
  val_p:=solve(identity(eq_p, t), {c[1], c[2]});
```

$$eq\_p := -3\,c_1\cos(3\,t) + 3\,c_2\sin(3\,t) = 3\sin(3\,t)$$

$$val\_p := \{c_2 = 1, c_1 = 0\}$$

A particular solution to this differential equation is given by

```
> yp:=subs(val_p, ytry);
```

$$yp := \cos(3\,t)$$

```
>
```

16: Page 187

Code the differential equation.

```
> deq:=diff(theta(t), t$2)-theta(t)=t*sin(t);
```

$$deq := \left(\frac{d^2}{dt^2}\theta(t)\right) - \theta(t) = t\sin(t)$$

The differential family of $t\sin(t)$ is $\{t\sin(t), t\cos(t), \sin(t), \cos(t)\}$. Moreover, since $\sin(t)$ is not a solution to the homogeneous equation, we use a trial solution of the form

$$y_{try}(t) = (b_0 + b_1\,t)\sin(3\,t) + (a_0 + a_1\,t)\cos(3\,t)$$

```
> ytry:=(b[0]+b[1]*t)*sin(t)+(a[0]+a[1]*t)*cos(t);
```

$$ytry := (b_0 + b_1\,t)\sin(t) + (a_0 + a_1\,t)\cos(t)$$

Substitute this trial solution into the differential equation and solve for a_0, a_1, b_0 and b_1.

```
> eq_p:=eval(subs(theta(t)=ytry, deq));
  pars:=solve(identity(eq_p, t), {a[0], a[1], b[0], b[1]});
```

$$eq\_p := 2\,b_1\cos(t) - 2\,(b_0 + b_1\,t)\sin(t) - 2\,a_1\sin(t) - 2\,(a_0 + a_1\,t)\cos(t) = t\sin(t)$$

$$pars := \left\{a_1 = 0, b_1 = \frac{-1}{2}, a_0 = \frac{-1}{2}, b_0 = 0\right\}$$

A particular solution to this differential equation is given by

```
> theta_p:=subs(pars, ytry);
```

$$theta\_p := -\frac{1}{2}\,t\sin(t) - \frac{1}{2}\cos(t)$$

```
>
```

25: Page 187

Code the differential equation.

```
> deq:=diff(y(t), t$2)+2*diff(y(t), t)+4*y(t)=111*exp(2*t)*cos(3*t);
```

$$deq := \left(\frac{d^2}{dt^2}\,\mathrm{y}(t)\right) + 2\left(\frac{d}{dt}\,\mathrm{y}(t)\right) + 4\,\mathrm{y}(t) = 111\,\mathbf{e}^{(2\,t)}\cos(3\,t)$$

To argue that $\mathbf{e}^{(2\,t)}\cos(3\,t)$ is not a solution to the corresponding homogeneous equation, we only need only look at $-\frac{b}{2\,a}$, where $a\,r^2 + b\,r + c = 0$ denotes the auxiliary equation.

$$-\frac{b}{2\,a} = -\frac{2}{2} = -1 \neq 2$$

We conclude that $\mathbf{e}^{(2\,t)}\cos(3\,t)$ is not a solution to the homogeneous equation. Moreover, the differential family of $\mathbf{e}^{(2\,t)}\cos(3\,t)$ is given by $\{\mathbf{e}^{(2\,t)}\cos(3\,t), \mathbf{e}^{(2\,t)}\sin(3\,t)\}$, so we can use a trial solution of the form

$$y_{try}(t) = c_1\,\mathbf{e}^{(2\,t)}\cos(3\,t) + c_2\,\mathbf{e}^{(2\,t)}\sin(3\,t)$$

```
> ytry:=c[1]*exp(2*t)*cos(3*t)+c[2]*exp(2*t)*sin(3*t);
```

$$ytry := c_1\,\mathbf{e}^{(2\,t)}\cos(3\,t) + c_2\,\mathbf{e}^{(2\,t)}\sin(3\,t)$$

Substitute this trial solution into the differential equation and solve for c_1 and c_2.

```
> eq_p:=eval(subs(y(t)=ytry, deq));
  pars:=solve(identity(eq_p, t), {c[1], c[2]});
```

$$eq\_p := 3\,c_1\,\mathbf{e}^{(2\,t)}\cos(3\,t) - 18\,c_1\,\mathbf{e}^{(2\,t)}\sin(3\,t) + 3\,c_2\,\mathbf{e}^{(2\,t)}\sin(3\,t) + 18\,c_2\,\mathbf{e}^{(2\,t)}\cos(3\,t) = 111\,\mathbf{e}^{(2\,t)}\cos(3\,t)$$

$$pars := \{c_1 = 1, c_2 = 6\}$$

A particular solution to this differential equation is given by

```
> yp:=subs(pars, ytry);
```

$$yp := \mathbf{e}^{(2\,t)}\cos(3\,t) + 6\,\mathbf{e}^{(2\,t)}\sin(3\,t)$$

```
>
```

28: Page 186

Since it is obvious that $\mathbf{e}^t$ is not a solution of the homogeneous equation

$$y'' + 3\,y' - 7\,y = 0$$

we can apply a trial solution of the form

$$y_{try}(t) = (c_0 + c_1\,t + c_2\,t^2 + c_3\,t^3 + c_4\,t^4)\,\mathbf{e}^t$$

in order to find a particular solution of the inhomogeneous equation

$$y'' + 3\,y' - 7\,y = t^4\,\mathbf{e}^t$$

31: Page 186

Code the differential equation.

```
> deq:=diff(y(t), t$2)+2*diff(y(t), t)+2*y(t)=8*t^3*exp(-t)*sin(t);
```

$$deq := \left(\frac{d^2}{dt^2}\,\mathrm{y}(t)\right) + 2\left(\frac{d}{dt}\,\mathrm{y}(t)\right) + 2\,\mathrm{y}(t) = 8\,t^3\,\mathbf{e}^{(-t)}\sin(t)$$

Observe that is this case $-\frac{b}{2\,a} = -1$. Hence, the simple technique used in Exersize 25 will not apply here. To find out whether or not $\mathbf{e}^{(-t)}\sin(t)$ is a solution to the corresponding homogeneous equation, we will actually have to find the eigenvalues.

```
> deqh:=lhs(deq)=0;
  aux_eq:=simplify(eval(subs(y(t)=exp(r*t), deqh))/exp(r*t));
  evals:=solve(aux_eq, r);
```

$$deqh := \left(\frac{d^2}{dt^2}\,\mathrm{y}(t)\right) + 2\left(\frac{d}{dt}\,\mathrm{y}(t)\right) + 2\,\mathrm{y}(t) = 0$$

$$aux\_eq := r^2 + 2\,r + 2 = 0$$

$$evals := -1 + I, -1 - I$$

Clearly, $\mathbf{e}^{(-t)}\sin(t)$ is a solution of the homogeneous equation, so we use a trial solution of the form

$$y_{try}(t) = t\,((b_0 + b_1\,t + b_2\,t^2 + b_3\,t^3)\,\mathbf{e}^{(-t)}\sin(t) + (a_0 + a_1\,t + a_2\,t^2 + a_3\,t^3)\,\mathbf{e}^{(-t)}\cos(t))$$

>

33: Page 186

Code the differential equation.

```
> deq:=diff(y(t), t$3)-diff(y(t), t$2)+y(t)=sin(t);
```

$$deq := \left(\frac{d^3}{dt^3}\,y(t)\right) - \left(\frac{d^2}{dt^2}\,y(t)\right) + y(t) = \sin(t)$$

Clearly, $\sin(t)$ is not a solution of the corresponding homogeneous equation, so we use a trial solution of the form

$$y_{try}(t) = c_1\sin(t) + c_2\cos(t)$$

```
> ytry:=c[1]*sin(t)+c[2]*cos(t);
```

$$ytry := c_1\sin(t) + c_2\cos(t)$$

Substitute this trial solution into the differential equation and solve for c_1 and c_2.

```
> eq_p:=eval(subs(y(t)=ytry, deq));
  val_p:=solve(identity(eq_p, t), {c[1], c[2]});
```

$$eq\_p := -c_1\cos(t) + c_2\sin(t) + 2\,c_1\sin(t) + 2\,c_2\cos(t) = \sin(t)$$

$$val\_p := \{c_2 = \frac{1}{5}, c_1 = \frac{2}{5}\}$$

A particular solution to this differential equation is given by

```
> yp:=subs(val_p, ytry);
```

$$yp := \frac{2}{5}\sin(t) + \frac{1}{5}\cos(t)$$

>